

$X_0(p)$

John Cremona

7/21/2014

Contents

$X_0(2)$: ramification data $(1^3, 1^2 + 1, 1^2 + 1)$ above $j = 0, 1728, \infty$:

```
R.<a,b,c> = QQ[]
```

```
S.<x> = R[]
```

```
j = (x+a)^3/x
```

```
eq = (j-1728).numerator() - (x+b)^2*(x+c)
```

```
eqs = eq.coefficients()
```

```
I = R.ideal(eqs)
```

```
I.variety()
```

```
[[a: -16, b: 8, c: -64], [a: 16, b: -8, c: 64]]
```

```
Zt.<t> = ZZ[]
```

```
j = (t+16)^3/t
```

```
print j.factor()
```

```
print (j-1728).factor()
```

```
t^-1 * (t + 16)^3
```

```
t^-1 * (t + 64) * (t - 8)^2
```

$X(2)$: degree = 6, ramification data $((1+1)^3, (1+1+1)^2, (1+1+1)^2)$ above $j = 0, 1728, \infty$:

```
R.<a1,a2,b1,b2,b3,c> = QQ[]
```

```
S.<x> = R[]
```

```
j = c*(x^2+a1*x+a2)^3 / (x*(x-1))^2
```

```
eq = (j-1728).numerator() - c*(x^3+b1*x^2+b2*x+b3)^2
```

```
R.ideal(eq.coefficients()).variety()
```

```
[[a2: 0, c: -6912, b1: -3/2, a1: -1, b3: 0, b2: 1/2], [a2: 1, c: 256, b1: -3/2, a1: -1, b3: 1, b2: -3/2]]
```

```
j = 256*(t^2-t+1)^3 / (t*(t-1))^2
```

```
print j.factor()
```

```
print (j-1728).factor()
```

```
2^8 * (t - 1)^-2 * t^-2 * (t^2 - t + 1)^3
```

$$2^6 * (t - 1)^{-2} * t^{-2} * (t - 2)^2 * (t + 1)^2 * (2*t - 1)^2$$

$X_0(3)$: degree 4, ramification data $(1^3 + 1, (1 + 1)^2, 1^3 + 1)$ above $j = 0, 1728, \infty$:

```
R.<a,b1,b2,c> = QQ[]
S.<x> = R[]
j = ((x+a)^3*(x+c))/ x
k = (x^2+b1*x+b2)^2 / x
eq = (j-k-1728)
R.ideal(eq.numerator().coefficients()).variety()
[{a: 3, c: 27, b1: 18, b2: -27}]
```

```
j = (t+3)^3 * (t+27) / t
print j.factor()
print (j-1728).factor()
t^-1 * (t + 27) * (t + 3)^3
t^-1 * (t^2 + 18*t - 27)^2
```

$X_0(5)$: degree 6, ramification data $((1 + 1)^3, (1 + 1)^2 + 1 + 1, 1^5 + 1)$ above $j = 0, 1728, \infty$:

```
R.<a1,a2,b1,b2,c1,c2> = QQ[]
S.<x> = R[]
j = (x^2+a1*x+a2)^3 / x
k = (x^2+b1*x+b2)^2 * (x^2+c1*x+c2) / x
eq = (j-k-1728)
R.ideal(eq.numerator().coefficients()).variety()
[{a2: 5, b1: 4, a1: 10, c1: 22, b2: -1, c2: 125}]
```

```
j = (t^2+10*t+5)^3 / t
print j.factor()
print (j-1728).factor()
t^-1 * (t^2 + 10*t + 5)^3
t^-1 * (t^2 + 22*t + 125) * (t^2 + 4*t - 1)^2
```

$X_0(7)$: degree 8, ramification data $((1 + 1)^3 + 1 + 1, (1 + 1 + 1 + 1)^2, 1^7 + 1)$ above $j = 0, 1728, \infty$:

```
R.<a1,a2,a3,a4,b1,b2,b3,b4> = QQ[]
S.<x> = R[]
j = (x^2+a1*x+a2)^3 * (x^2+a3*x+a4)/ x
k = (x^4+b1*x^3+b2*x^2+b3*x+b4)^2 / x
eq = (j-k-1728)
R.ideal(eq.numerator().coefficients()).variety()
```

```
j = (t^2+5*t+1)^3 * (t^2+13*t+49) / t
k = j-1728
print j.factor()
print k.factor()
t^-1 * (t^2 + 13*t + 49) * (t^2 + 5*t + 1)^3
t^-1 * (t^4 + 14*t^3 + 63*t^2 + 70*t - 7)^2
```